

Easy-PCB Solder Reflow Oven

Ben Lorenzetti & Adam Meece

Features

- Insulated aluminum frame with 6"x6"x10" interior chamber
- 400 Watt nichrome heating element with door safety switch
- 120 VAC power supply with surge and short protection
- Forced convection fan driven by bipolar stepper motor
- Type K thermocouple for temperature feedback loop
- PIC18F14K50 based control
- Convenient 7-segment display and pushbutton user interface

Introduction


Easy-PCB is a small, desktop convection oven for DIY reflow soldering at home.

As integrated circuits continue to become smaller and faster, many μ Controllers, FPGAs, and other ICs are no longer available in DIP form for breadboard prototyping or iron soldering. Similar to the shift away from PC parallel ports in the early 2000s, the shift away from DIPs means modern electronics hobbyists need more complicated equipment than their predecessors.

Currently, commercial solder reflow stations are available for soldering fine pitch, SMD, and BGA components. More recently, several hackers have created open source designs for inexpensive reflow ovens, using converted kitchen toaster ovens.

Easy-PCB is better because it is a convection oven; controlled air flow prevents parts from being blown out of position and convection heating obviates the uneven

heat absorption seen in infrared radiation based systems. Furthermore, *Easy-PCB* was designed from the ground up for soldering, so the heating elements and oven frame were optimized alongside the temperature controller for a typical solder reflow profile.



The *Easy-PCB* Control System: not just a controller bolted onto a disjoint oven.

Easy-PCB allows hobbyists to produce reflow temperature profiles with a high degree of accuracy and repeatability. Alongside freely available PCB design programs like Eagle and low-quantity fabrication services like OSH Park, *Easy-PCB* makes modern integrated components, such as digital CMOS cameras, within the design space of hobbyists and hackers.

control-results.pdf


Faithful Temperature Profile Control

Contents


Features	1
Introduction	1
1 User Interface	2
2 Plant Design	2
2.1 Temperature Objectives	2
2.2 Thermodynamics Theory	3
2.3 Thermodynamic Design	4
2.4 Design Iteration & SPICE Modeling .	8
2.5 Plant Bill of Materials	9
3 User Interface Design	9
3.1 Separate, Front-Facing PCB	9
3.2 Door E-Stop	9
3.3 Bill of Materials	10
4 Temperature Feedback Design	10
4.1 Thermocouple & Amplifier	10
4.2 Bill of Materials	11
5 Motor Driver Design	11
5.1 Push-Pull Driver	11
5.2 Driver SPICE Simulation	12
5.3 Bill of Materials	13
6 Power Circuit Design	13
6.1 PCB Trace Width	13
7 Controller Design	13
A SPICE Netlists	13
A.1 Thermodynamic Model	13
A.2 Bipolar Motor Driver	13

1 User Interface

Easy-PCB presents temperature and time information on a four digit LED display. After pressing the green ‘start’ button, the current temperature and process runtime are alternately displayed every second.





The red ‘stop’ button can be pressed at anytime to quit the current process or reset the microcontroller.



Pushbutton Interface

Set point programming with the 4 nav keys is used to enter a time-temperature profile. Use the ‘left’ and ‘right’ keys to move between set points and the ‘up’ and ‘down’ keys to edit each set point value.



Set Point Programming

A total of 128 set points can be entered. The first set point is always $(0s, 20^\circ C)$, and the process will terminate at the next occurrence of 20° in a point. Temperatures can take any value in the range $20^\circ C < T < 275^\circ C = 527^\circ F$. The time between two points must be between $0 \leq t < 255$ seconds.

2 Plant Design

2.1 Temperature Objectives

A typical solder reflow profile is shown in figure 1. The optimal shape and peak temperature depend on the components and, more significantly, on the use of leaded solder.

$$T_p \approx T_c = \begin{cases} 220 - 235^\circ C & \text{Leaded solder} \\ 245 - 260^\circ C & \text{Pb-free} \end{cases}$$

$$t_p = \begin{cases} 20s & \text{Leaded solder} \\ 30s & \text{Pb-free} \end{cases}$$

Easy-PCB was designed from the ground up for faithfully producing these reflow profiles. This design process for the thermodynamic system is detailed below.




Figure 1: Standard Solder Reflow Profile, per IPC/JEDEC STD-1-020D.1

2.2 Thermodynamics Theory

Heat is everyone's least favorite form of energy because it tends to dissipate everywhere and is difficult to transform into useful work. Nevertheless, it is a form of energy so it is measured in joules.

$$Q := \text{Heat [Joules]}$$

From a microscopic view, heat is the kinetic energy of gas molecules or the vibrational energy of bonds in solids. In both cases, summed over every molecule, every bond, and every vibrational mode in the system.




Figure 2: Heat Energy on a Molecular Scale

Temperature is a potential field that measures an object's tendency to give up or absorb heat. On average, heat always flows from regions of hotter temperature to lower temperature. Temperature is usually measured in Kelvin or Celsius, which are related by

$$T_{\text{Celsius}} = T_{\text{Kelvin}} + 273.16.$$

Another way to think of temperature is a measure of heat density that also incorporates the types of bonds (or lack thereof) in which heat energy is stored. For example, a glacier contains more heat than a pot of boiling water, but has a much lower heat density.

The relationship between temperature and heat stored in an object is given by

$$dQ = mc_p dT, \quad (1)$$

where c_p is the specific heat of the material with units $\left[\frac{\text{kJ}}{\text{kg}\cdot\text{K}} \right]$.

So heat flows spontaneously to reduce temperature gradients, but how does heat flow and at what rate? There are three mechanisms of heat transfer: *conduction*, *convection*, and *radiation*.

Conduction of heat is caused by molecular interactions, such as the push-pull of nearby atomic bonds

in solids or elastic collisions between molecules of air.

The empirical relation for conduction is Fourier's Law, which says the rate of heat transfer due to conduction is proportional to both the difference in temperature and the media:

$$\vec{q} = k \vec{\nabla} T$$

where \vec{q} is the rate of heat transfer per unit surface area and k is the thermal conductivity of the medium with units $[\frac{W}{m^2 K}]$.

In integral form, Fourier's Law is

$$\frac{\delta Q}{\delta t} = k \oint_S \vec{\nabla} T * d\vec{A}.$$

And, in the one dimensional case,

$$\frac{\delta Q}{\delta t} = -kA \frac{\delta T}{\delta x}. \quad (2)$$

Convection is the second mechanism of heat transfer, involving the bulk movement of particles driven by diffusion.

Every air molecule is moving in a random direction with random kinetic energy, but a region of air with higher average kinetic energy (temperature) will see more molecules leaving than entering on average, because those leaving are moving faster than those entering.

Note that in convection, no molecules gain or lose energy; molecules of different energy simply trade places.

If a fan is used to apply work to a gas, the rate of convection increases and then Newtons' Law of Cooling can predict the rate of convection in gas near a solid surface of different temperature.

$$\frac{dQ}{dt} = hA\Delta T \quad (3)$$

where h is the heat transfer coefficient with units $[\frac{W}{m^2 K}]$.

Radiation, the third mechanism of heat transfer, is more familiar to electrical engineers. In any material, electrons have discrete amounts of energy based on the wave patterns that can exist for the atomic geometry. When an electron spontaneously falls to a less energetic patten, that energy is emitted as a photon of light.

The classic, physics-history example is black body radiation, when a metal is heated to high temperature and glows. A more modern example is the LED.

The empirical relation for radiation is the Stefan-Boltzmann Law, which says that the total energy radiated, over all wavelengths and per unit surface

area, is proportional to the fourth power of the body's temperature.

$$j^* = \sigma T^4 \quad (4)$$

where σ is a proportionality constant derived from other constants of nature.

Mechanisms & Media Usually one mechanism of heat transfer is dominant in a particular medium, so much so that we can ignore the other two. Typically this means conduction in solids, convection in fluids or gas, and radiation in space. However, sometimes the picture is more complicated including two cases in an oven:

1. At the boundary of metal and air, where conduction and convection are both significant.
2. In a thermocouple loop, where thermal diffusion (convection) of electrons is useful despite conduction still being dominant.

2.3 Thermodynamic Design

Most microwaves draw between 600 and 1200 Watts of power. An IR toaster oven may draw 1500 Watts into its lamps. How much power does *Easy-PCB* need to produce the required temperature profiles?

Our approach for answering this question is to browse available parts on McMaster.com, imagine how they would be assembled into an oven, and then model their thermodynamic response using SPICE.

A first conceptual design of *Easy-PCB* is shown in figure 3.




Figure 3: Concept for Thermodynamic System

Transmission through Metals & Insulation

In the concept, the oven chamber is isolated from the outside with walls of mineral wool insulation and

aluminum. If the wall is relatively thin compared to the square root of its surface area, then its heat transfer can be modeled as a 1-D transmission line, like shown in figure 4.




Figure 4: Heat Conduction in 1-D

Assuming conduction is the dominant mechanism of heat transfer and using Fourier's Law in 1-D, (equation 2) the rate of heat passing through a thin slice is

$$\frac{\delta Q}{\delta t} = -kA \frac{\delta T}{\delta x}$$

And, from equation 1, the heat capacity of a thin slice of the metal sheath is

$$dQ = (\rho Adx)c_p dT$$

Rewriting these equations to look like current-voltage relationships gives

$$\delta T = R_x \frac{\delta Q}{\delta t}, \quad R_x = \frac{\delta x}{kA} \quad (5)$$

and

$$\frac{\delta Q}{\delta t} = C_x \frac{\delta T}{\delta t}, \quad C_x = \rho A c_p * \delta x \quad (6)$$

Values for thermal conductivity k , specific heat capacity c_p , and density ρ of different materials are listed in table 1.

Long metal bars and rods can also be modeled with equations 5 and 6 if the heat loss along the length of the bar is assumed to be negligible compared to heat transferred at the end faces.




Figure 5: 1-D Conduction in Both Directions

For metal components where heat is injected at the center of the part and dissipates symmetrically in either direction, such as the support bar shown in figure 5 or the metal sheath encasing the heating element, the surface area can be doubled so that the transmission line only needs to be integrated in the positive direction.

$$\left\{ \begin{array}{lcl} A_x & \Rightarrow & 2A_x \\ R_x & \Rightarrow & R_x/2 \\ C_x & \Rightarrow & 2C_x \\ \{x| -\frac{w}{2} \leq x \leq \frac{w}{2}\} & \Rightarrow & \{x| 0 \leq x \leq \frac{w}{2}\} \end{array} \right. \quad (7)$$

Resistivity of Metal-Air Interface

In the bulk volume of air, convection is the dominant form of heat transfer and collisions are relatively rare. However, where the air meets a solid surface, the net convection must go to zero because there can be no net movement of air molecules into the solid. Furthermore, every air molecule that reaches the surface experiences a collision so there is 100% conduction at the surface.

For some distance away from the surface, there is a larger than normal chance of head-on collisions and slower rates of diffusion due to the nearby wall. In this region, both conduction and convection may be significant. To analyze this transition region, we will start at the surface where there is 100% conduction because that is the mechanism we have equations for.

From the first law of thermodynamics, the rate of heat being stored in a differential volume is equal to the rate of heat entering less the rate of heat leaving the volume element.

$$\frac{\delta Q_{\text{stored}}}{\delta t} = \frac{\delta Q_{\text{in}}}{\delta t} - \frac{\delta Q_{\text{out}}}{\delta t}$$

Looking back at figure 4, the differential volume element is a plane, so

$$\frac{\delta Q_{\text{stored}}}{\delta t} = \frac{\delta Q_x}{\delta t} - \frac{\delta Q_{x+\delta x}}{\delta t}.$$

Substituting in equation 1 for stored heat capacity and equation 2 for conduction through the element yields a transport equation.

Table 1: Thermal Conductivity, Specific Heat, and Density of Selected Materials

Material	Linear Range ($^{\circ}C$)	$k \left(\frac{W}{m*K} \right)$	$c_p \left(\frac{kJ}{kg*K} \right)$	$\rho (kg/m^3)$
Type 6061 Aluminum	25	167	0.896	2700
Type 304 Stainless Steel	0–100	16.2	0.5	8000
Cast Iron	25	55.0	0.46	6800–7800
1.0 K-factor Insulation	-	0.1442	-	-
0.23 K-factor Insulation	-	0.0332	-	-
Nichrome NiCr C Alloy	-	0.450	0.135	8244
Steatite L5 Ceramic	-	2.9	0.92	2710

$$mc_p \frac{\delta T}{\delta t} = kA \frac{\delta T}{\delta x} \quad (8)$$

This cannot be solved analytically unless space and time are independent. Sorry Albert.

$$T(x, t) = T_x(x) * T_t(t) \quad (9)$$

Applying the chain rule to equation 8

$$mc_p \frac{\delta T_t(t)}{\delta t} T_x(x) = kA \frac{\delta T_x(x)}{\delta x} T_t(t)$$

and separating variables gives

$$mc_p \frac{\delta T_t(t)}{\delta t} * \frac{1}{T_t(t)} = kA \frac{\delta T_x(x)}{\delta x} * \frac{1}{T_x(x)}.$$

The only way for the equality to be true for all space and time is if both sides are equal to a common constant.

$$\alpha = mc_p \frac{\delta T_t(t)}{\delta t} * \frac{1}{T_t(t)}$$

$$\alpha = kA \frac{\delta T_x(x)}{\delta x} * \frac{1}{T_x(x)}$$

Each of these can be solved by separation of variables.

$$\frac{\alpha}{kA} \delta x = \frac{\delta T_x(x)}{T_x(x)}$$

$$\frac{\alpha}{kA} \int \delta x = \int \frac{1}{T_x(x)} * \delta T_x(x)$$

$$\frac{\alpha x}{kA} = \ln(T_x(x)) + \beta_1$$

$$T_x(x) = \beta'_1 e^{\alpha x / kA}$$

Similarly for time,

$$T_t(t) = \beta'_2 e^{\alpha t / mc_p}.$$

Recombining the independent components gives

$$\Delta T(x, t) = \beta'_1 \beta'_2 e^{\alpha x / kA + \alpha t / mc_p}$$

The boundary conditions are that at $t = 0$ and $x = 0$, the temperature difference is the initial temperature difference, and at $t \rightarrow \infty$ and $x \rightarrow \infty$, the temperature difference dissipates to zero.

$$\Delta T(x, t) = \Delta T_0 e^{-\alpha x / kA - \alpha t / mc_p}$$

Now, for the sake of physical intuition we can write

$$\Delta T(x, t) = \Delta T_0 e^{-(x/\delta + t/\tau)}, \quad (10)$$

where δ is a ‘skin depth’ over which the spacial gradient decays to $1/e$ its initial value and τ is a time constant related to the skin depth by

$$\tau = \frac{mc_p}{kA} \delta. \quad (11)$$

Make some arguments and plot the temperature and heat transfer curves. Assume steady state and somehow that active cooling makes it all valid.

Applying Fourier’s law to the temperature profile found in equation 10, gives us the rate of heat transfer in the air due to conduction.

$$\frac{dQ}{dt} = -kA \frac{dT}{dx} = \frac{kA}{\delta} \Delta T_0 e^{-(x/\delta + t/\tau)}$$

Again assuming steady state, the rate of heat transfer at any distance from the wall is equal to the rate of conduction at the wall.

$$\frac{dQ}{dt} = \frac{kA}{\delta} \Delta T_0 = h_c A \Delta T_0 \quad (12)$$

This is the information we were after and, ironically, it turns out this is Newton’s Law of Cooling (equation 3) where $h_c = k/\delta$. Since we do not have any way of predicting the skin depth except by experimentation, some common values for the convective heat transfer coefficient h_c are listed in table 2.

Equation 12 can be rewritten to look like Ohms’ law for SPICE analysis.

$$\Delta T = R_c \frac{dQ}{dt}, \quad R_c = \frac{1}{h_c A} = \frac{\delta}{kA} \quad (13)$$

Table 2: Typical Convective Heat Transfer Coefficients

Fluid	Conv.	$h_c \left[\frac{W}{m^2 * K} \right]$
Gases, & dry vapors	free	0.5 – 1000
	forced	10 – 1000
Water & liquids	free	50 – 3000
	forced	50 – 10000
Boiling Water	-	3 – 100
Condensing H_2O Vapor	-	5 – 100

Table 3: Thermal Properties of Air

$T [^{\circ}C]$	$\rho \left[\frac{kg}{m^3} \right]$	$c_p \left[\frac{kJ}{kg * K} \right]$	$k \left[\frac{W}{m * K} \right]$
20	1.205	1.005	0.0257
40	1.127	1.005	0.0271
60	1.067	1.009	0.0285
80	1.000	1.009	0.0299
100	0.946	1.009	0.0314
120	0.898	1.013	0.0328
140	0.854	1.013	0.0343
160	0.815	1.017	0.0358
180	0.779	1.022	0.0372
200	0.746	1.026	0.0386
250	0.675	1.034	0.0421
300	0.616	1.047	0.0454

Heat Capacity of Chamber Air

The heat capacity of the air in the chamber cannot be analyzed with constant coefficient assumptions.

$$dQ = m(T)c_p(T)dt \quad (14)$$

From table 3, we can see that the specific heat of air does not vary much over the temperature range of the oven; less than 0.3% from $c_{p(20^{\circ}C)}$ over the range $20 - 300^{\circ}C$. However, the density and mass of air in the chamber may change significantly.

Assuming that the air in the chamber behaves like an ideal gas and is made up of the normal percentages of N_2 , O_2 , CO_2 , etc., then mass and temperature are related by

$$PV = \frac{m}{M}RT \Rightarrow m(T) = \frac{MPV}{R}T$$

where the ideal gas constant is

$$R = 5.00745 \left[\frac{\text{atm} * \text{in}^3}{\text{mol} * \text{K}} \right]$$

and the molar mass of air is

$$M = 28.97 \left[\frac{g}{\text{mol}} \right]$$

For the first oven concept in figure 3, the volume of the chamber is fixed in natural units of in^3 and there are likely small leaks that let the chamber pressure equilibrate to atmospheric pressure 1 atm.

Substituting $m(T)$ into equation 14, converting from absolute Kelvin scale to Celsius, and manipulating to look like a current–voltage relationship results in

$$\frac{dQ}{dt} = C_c(T) \frac{dT}{dt} \quad (15)$$

where

$$C_c(T) = \frac{c_{p(20^{\circ}C)} MPV}{R(T + 273.16^{\circ}C)}. \quad (16)$$

Equivalently,

$$C_c(T) = \frac{(5.8143 \left[\frac{J}{in^3} \right]) * V}{T + 273.16^{\circ}C}.$$

Variable Capacitor for Chamber Air

Unfortunately, SPICE does not have a primitive circuit element for a variable capacitor. Before simulating and iterating the oven design, we must build a SPICE subcircuit to model heat capacity of ideal gasses. The subcircuit black box must have two terminals with I–V relation

$$i_c = C(v_c) \frac{dv_c}{dt},$$

or, alternatively,

$$\int i_c dt = \int C(v_c) dv_c. \quad (17)$$

The left side of equation 17 is an integrator, which can be implemented with a buffer and a capacitor. For the SPICE circuit shown in figure 6, the integrated current is proportional to the voltage of the ‘int’ node.

$$\int i_c dt = C_{int} * v_{int}$$




Figure 6: Voltage Variable Capacitor in SPICE

Table 4: Nichrome Wire Properties (NiCr C)

Composition	59.2%Ni	16%Cr	23.5%Fe	1.3%
Resistivity (ρ)	$675 * \pi$	$\left[\frac{\Omega \cdot \text{mil}^2}{\text{ft}} \right]$		
Melting Pt.	1350	${}^{\circ}\text{C}$		
Temp. [${}^{\circ}\text{C}$]	20	93	204	300
% ρ Increase	0	1.7	3.5	5

For the right side of equation 17, it can be solved for the variable heat capacitance of an ideal gas found in equation 16.

$$\begin{aligned} \int C(v_c) dv_c &= \frac{c_p MPV}{R} \int \frac{dv_c}{v_c + 273.16} \\ &= \frac{c_p MPV}{R} \ln(v_c + 273.16) + A \end{aligned}$$

Recombining the left and right halves gives

$$C_{int} v_{int} = \frac{c_p MPV}{R} \ln(v_c + 273.16) + A,$$

which can be solved to find a nonlinear expression for a dependent source v_c .

Solving...

$$\frac{RC_{int}}{c_p MPV} * v_{int} = \ln(v_c + 273.16) + A$$

$$v_c = A' e^{v_{int}/B} - 273.16$$

Finally, applying boundary conditions $v_c(0) = 0$ and $v_{int}(0) = 0$ gives

$$v_c = 273.16 \left(e^{v_{int}/B} - 1 \right) \quad (18)$$

where constant B is

$$\begin{aligned} B &= \frac{c_p MP}{R} * V * C_{int} \\ &= (5.8413 \left[\frac{J}{in^3} \right]) * V * C_{int} \end{aligned} \quad (19)$$

and V is the constant volume of air.

2.4 Design Iteration & SPICE Modeling

Many designs with different shapes, sizes, metals, insulators, heating elements, and wattage levels were modeled in SPICE for their thermodynamic response to a unit step signal from the power relay. The primary parameters of interest was the heating rate of the air in the chamber.

The SPICE models were based on the equations from section 2.3; material properties from




Figure 7: Aluminum Frame Concept




Figure 8: Heating Element Concept

tables 1, 2, 3, and 4; and geometry of the design. Unitwise, 1 Volt = 1°C , 1 Ampere = 1 Watt, and 1 Coulomb = 1 Joule.

Eventually a design using 28 ft of 45 mil nichrome wire inside a 8" x 8" x 12" aluminum frame was selected.

The shape of the aluminum frame is shown in figure 7. The inside of the frame is lined with 1" thick mineral wool insulation, leaving an inside chamber with dimensions 6" x 6" x 10". A fan is mounted in the back.

Inside the frame, the nichrome wire is mounted using high temperature ceramic spacers for electrical insulation. Structurally, the nichrome element is tensioned by two size 10 steel bolts spanning the width of the chamber, as shown in figure 8. Direct contact between the nichrome heating element and the chamber air is critical for fast heating rates.




Figure 10: Simulated Temperature Response to a Unit Step

The thermodynamic model is shown in figure 9.
DISCUSSION/MATH OF MODEL VALUES

The circuit netlist passed to SPICE is listed in section A.1. The circuit was modeled for its transient response if the heating element were connected to 120 VAC power for 100 second. The results are shown in figure 10.

2.5 Plant Bill of Materials

3 User Interface Design

3.1 Separate, Front-Facing PCB

The user interface is designed to be assembled on a separate PCB from the main controller board. This separation allows the high power components to be mounted on the back of the oven—far away from the user. Thus, the user interface PCB is mounted near the door and has no voltages greater than 12 V for safety. Everything on the front PCB is shown in blue in *Easy-PCB* block diagram.




Figure 11: Hardware Debounced Pushbuttons

The *Easy-PCB* Control System: not just a controller bolted onto a disjoint oven.

The user interface board contains a 4-digit, 7-segment, common-cathode display and a matching common-cathode driver. The display driver is connected to the controller board via 4 wire serial peripheral interface (SPI). From the MAX7219 datasheet, the worst case current draw by the display driver is

$$I_{LED \text{ Display}} < 200 \text{ mA.} \quad (20)$$

There are 6 active low pushbuttons on the front facing board with 6 corresponding low pass filters on the controller board. The pushbutton circuit is shown in figure 11.

Finally, there is a safety switch that is compressed when the door is closed.

3.2 Door E-Stop

The safety switch on the front-facing PCB board is form A: normally open and active closed. The current through this door switch directly drives a form A (normally open) safety relay on the controller board. The safety switch circuit is shown in figure 12. The flyback diode should be rated for currents greater than the saturation current of the relay coil.

$$I_{\text{Diode Rating}} > I_{\text{Relay Sat.}} = \frac{V_{\text{DC Coil Rating}}}{R_{\text{coil}}} \quad (21)$$

For the Omron Safety G4W relays, the saturation current is

$$I_{\text{Relay Sat.}} = 66.7 \text{ mA} \quad \text{at } 12 \text{ V} \quad (22)$$




Figure 9: Thermodynamic SPICE Model Circuit Diagram





Figure 12: Safety Switch Safety Circuit

3.3 Bill of Materials

4 Temperature Feedback Design

To develop accurate temperature control, real time temperature feedback is necessary. The temperature is too hot for thermisters or temperature ICs, so old school thermocouples must be used.



4.1 Thermocouple & Amplifier

The Seebeck voltage is the electromotive force that develops due to temperature gradient in a conductive loop made of two different metals. It is described by

$$E_S = \int_{T_R}^T \alpha_{A,B} dT, \quad (23)$$

where T and T_R are the temperatures of the hot and cold junctions and $\alpha_{A,B}$ is the Seebeck coefficient between materials A and B .

A type-K thermocouple was selected because they are readily available and resist oxidation at temperatures up to $1260^\circ C$. From the 'Manual on the use of Thermocouples in Temperature Measurement' (STP 470A), the nominal Seebeck coefficients for type-K thermocouples is

$$\alpha_{KP,KN} = \begin{cases} 0^\circ C & 39.4 \mu V/\text{ }^\circ C \\ 200^\circ C & 40.0 \mu V/\text{ }^\circ C \\ 400^\circ C & 42.3 \mu V/\text{ }^\circ C \end{cases}. \quad (24)$$




Figure 13: Thermocouple Amplifier

The positive wire KP is one of the following alloys: Chromel, Tophel, T-1, or ThermoKanthal KP. The negative wire KN is Alumel, Nial, T-2, or ThermoKanthal KN.

The thermoelectric potential in a K loop is very small; for a temperature difference of $T - T_R = 500^\circ C$, the voltage developed is only on the order of $E_S = 20 \text{ mV}$. Consequently, the signal needs to be amplified before it can be read by an ADC. The op amp circuit shown in figure 13 can be used.

Five volt microcontrollers usually have ADC ranges of 0–3.3 V. To scale 20 mV to 3.3 V and 0 V to 0 V, the feedback resistors should be

$$\frac{R_{f2}}{R_{f1}} = \frac{V_{\text{ADC}(\text{max})}}{V_{\text{K}(\text{max})}} = \frac{3.3 \text{ V}}{20 \text{ mV}}$$

$$R_{f2} = 165 * R_{f1} \quad (25)$$

4.2 Bill of Materials

5 Motor Driver Design

The combination of needing motors and relays and already being plugged in to a 120 VAC source led to the decision to create a 12 V power rail. The 12 V rail will only be used by components that are not particularly affected by a noisy signal, so this rail does not need active regulation.

With a 12 V rail in mind, there are many more choices for motors to drive the fan. A bipolar stepper motor was chosen because of its open speed control in low torque applications.




Table 5: Step Polarities for 44M100D2B Motor

CW	Step	Red	Gry	Yel	Blk	CCW
	1	+	-	+	-	^
	2	+	-	-	+	
	3	-	+	-	+	
v	4	-	+	+	-	

Figure 14: Bipolar Stepper Motor

The 44M100D2B motor by Portescap rotates 3.6° per change in polarity—i.e. 100 step increments, according to the polarities in table 5.

5.1 Push-Pull Driver

The driving circuit for the bipolar motor must be capable of pulling each winding lead to either side of the supply. A good implementation is with push-pull BJTs because no power is lost across the collectors when the output voltage matches the input. The driver circuit in figure 15 has 4 power transistors, Q_{1-4} , arranged in push-pull configurations.

A microcontroller can only output up to 5V, with limited current, so Q_{5-9} are needed to help bias the push-pull transistors' inputs.

It is bad news Bearcats if the wrong combinations of push-pull BJTs come on simultaneously. To protect from programmer error, Q_5 and Q_6 are arranged as a differential amplifier such that opposite transistors are biased on together.

The differential amplifier is biased by a constant current source, made from Q_{7-9} . If the current source is turned off, the push-pull inputs on either side of the winding will both rise to V_{CC} . Soon, the outputs will match and no current will flow through the winding, push-pull transistors, or the biasing transistors; i.e. the bipolar motor can be turned off to save power.

To design the driver, the electric properties of the motor windings and symmetric NPN/PNP BJTs are needed. The resistor networks should be chosen




Figure 15: Push-Pull Driver Circuit for a Bipolar Motor




Figure 16: Simulated Voltages across One Driver

such that switching V_{in} from 0 to 5 V causes Q_5 and Q_6 to alternate cutoff and saturation.

5.2 Driver SPICE Simulation

The circuit in figure 15 was simulated in SPICE with the net listed in section A.2. Five volt step functions were applied to both V_{on} and V_{in} to simulate the microcontroller. The results in figure 16 and figure 17 show how motor loads the power supply.




Figure 17: Simulated Current through One Driver

Table 6: 44M100D2B Electrical Specifications

	Bipolar
DC Operating Voltage	12V
Resistance per Winding	70Ω
Inductance per Winding	$35mH$
Rated Current per Phase	$12V/70\Omega = 0.17A$

Table 7: Thur-hole BJTs and Diodes

Label	Man. Part No.	Manufacturer	Doping	h_{fe}	$I_{C(max)}$ or I_o
Q_1, Q_2	MPSA06-AP	Micro Commercial Co	NPN	100	500mA
Q_3, Q_4	MPSA56-AP	Micro Commercial Co	PNP	100	500mA
Q_{5-9}	2N3904-AP	Micro Commercial Co	NPN	100	200mA
D_{1-4}	1N4007-TP	Micro Commercial Co	Standard PN	N/A	1A

5.3 Bill of Materials

6 Power Circuit Design

6.1 PCB Trace Width

The minimum width and clearance for high current, high voltage traces on a PCB can be calculated using

$$I = 0.048[A] * \left(\frac{\Delta T}{[{}^{\circ}C]} \right)^{0.44} * \left(\frac{A}{[mil^2]} \right)^{0.725} \quad (26)$$

and

$$\text{Clearance} = 0.023[in] + \frac{0.0002[in]}{[V]} * V_{\text{peak}} \quad (27)$$

from the design standard for PCB trace width ANSI/IPC-2221. Note that equation 26 is only for exterior traces.

For $10 A$, $120 VAC$ power with a maximum temperature rise of $10 {}^{\circ}C$ and standard copper thickness $1 oz = 1.4 mil$, the minimum trace width is $0.278 in$ and minimum clearance is $0.057 in$.

7 Controller Design

A SPICE Netlists

A.1 Thermodynamic Model

A.2 Bipolar Motor Driver

```

*** heat-model.cir ***
; Nodes: gnd, atm, bulkAir, interiorFrame, exteriorFrame, elementSurface,
;         ceramicSurface, bulkBolt, boltRadiators
;
; Nichrome Wire Heating Element Parameters:
.param Diameter=45      ; [0.001 in]
.param Length=336        ; [in]
.param Relectric={225*Length/(Diameter*Diameter)}      ; [Ohms]
.param wattage={(120*120)/Relectric}                   ; [Joules]
.param Volume={(1.287e-11)*Diameter*Diameter*Length} ; [m^3]
.param SurfaceArea={(2.027e-6)*Diameter*Length}       ; [m^2]
.param Celement={(1.113e6)*Volume}                     ; [J/K]
.param Relement={0.1/SurfaceArea}                      ; [K/W]
;
; Frame Parameters:
.param chamberVolume=360
.param Rinsulation=3.8
.param frameWidth=4.76mm
.param Rxframe=0.0181
.param Cxframe=798.3k
.param RexteriorFrame=6.06
;
; Ceramic Insulators and Bolt Mount Parameters:
.param Rparasitic={0.1*Relement}
.param ceramicThickness=2.35mm
.param Rxc=153.6
.param Cxc=5.598k
.param Cbolt=28.4
.param boltTlength=25.4mm
.param Rxb=843.7
.param Cxb=73.16
.param Rsink={RexteriorFrame/100}

Vroomtemp atm gnd DC 20
Ielement atm elementSurfacePrime PULSE(0 {wattage} 1s 1ns 1ns 100s 100s)
Vflowmeter elementSurfacePrime elementSurface DC 0
; (init_value, pulse_value, delay_t, rise_t, fall_t, pulse_width, period)

Cnichrome elementSurface atm {Celement}
Rnichrome elementSurface bulkAir {Relement}
Xairheatcapacity bulkAir atm varCap VOLUME={chamberVolume}
Rinsulate bulkAir interiorFrame {Rinsulation}
Ualuminumframe interiorFrame exteriorFrame gnd frameModel L={frameWidth}
Rframe exteriorFrame atm {RexteriorFrame}
Rparasitics elementSurface ceramicSurface {Rparasitic}
UceramicSpacers ceramicSurface bulkBolt gnd ceramicModel L={ceramicThickness}
CbulkBolt bulkBolt atm {Cbolt}
Ubolt bulkBolt boltEnd gnd boltTmodel L={boltTlength}
Rboltsink boltEnd atm {Rsink}

.MODEL frameModel URC RPERL={Rxframe} CPERL={Cxframe}
.MODEL ceramicModel URC RPERL={Rxc} CPERL={Cxc}
.MODEL boltTmodel URC RPERL={Rxb} CPERL={Cxc}
.subckt varCap n1 n2 VOLUME=360 ; inches squared
B1 n1 int2 V=273.16*(exp((v(int))/(5.8413*{VOLUME}))-1)
Vammeter int2 n2 0
F1 gnd int Vammeter 1
C1 int gnd 1
.ends varCap

.control
tran 0.1s 100s; tstep, tstop
plot v(elementSurface) v(bulkAir) i(Vflowmeter) v(exteriorFrame) v(boltEnd)

```

```
*** motor-driver.cir ***
```

```
Vcc cc gnd dc 12 ac 0
Q1 cc ppb1 out1 npn1
Q2 cc ppb2 out2 npn1
Q3 gnd ppb1 out1 pnp1
Q4 gnd ppb2 out2 pnp1
D1 out1 cc d1
D2 out2 cc d1
D3 gnd out1 d1
D4 gnd out2 d1
Lwinding out1 split 51.7mH
Rwinding split out2 52.4
Rc1 cc ppb1 450
Rc2 cc ppb2 450
Vin1 in gnd DC 0V PULSE(0V 5V 0ms 1us 1us 10ms 20mS)
R1 in b5 1000
R2 b5 gnd 1000
R3 b6 cc 4000
R4 b6 gnd 800
Q5 ppb1 b5 ifork npn1
Q6 ppb2 b6 ifork npn1
Von on gnd DC 0V PULSE(0V 5V 40ms 1us 1us 40ms 80ms)
Q7 cc on e7 npn1
Rref e7 imirror 146.5
Q8 imirror imirror e8 npn1
Q9 ifork imirror e9 npn1
Re8 e8 gnd 10
Re9 e9 gnd 10

.model npn1 NPN (BF=100 CJC=3pf CJE=5pf IS=1E-16 VAF=100 NF=1)
.model pnp1 PNP (BF=100 CJC=3pf CJE=5pf IS=1E-16 VAF=100 NF=1)
.model d1 D (BV=50 IS=1E-13)

.control
tran 1E-5 8E-2
plot v(in) v(b5) v(b6) v(out1) v(out2)
plot ((v(split)-v(out2))/52.5) (-i(Vcc))
```